Quantum-Mechanical Diffraction and Interference Patterns for Sharp-Edged Slits

W. P. Healy
Department of Mathematics
RMIT University
liam.healy@rmit.edu.au

R. A. Samandra
Department of Mathematics
RMIT University

Abstract

Quantum-mechanical diffraction and interference patterns associated with one-dimensional motion of a non-relativistic particle are obtained by using Maple. The normalized probability density for the particle's position after passage of the wave function through one or more sharp-edged slits involves combinations of Fresnel cosine and sine integrals and can readily be plotted. The Aharonov-Bohm phase shift caused by a magnetic flux string located between two slits is also considered. The periodicity of the Aharonov-Bohm effect is illustrated in a Maple animation in which the interference pattern varies with the value of the confined magnetic flux. The classical limit is demonstrated in a separate animation in which Planck's constant is made to approach zero and it is shown that the Aharonov-Bohm effect vanishes in this limit.

1 Introduction

The use of Fresnel cosine and sine integrals has long been established in the classical theory of the diffraction and interference of light due to rectangular apertures and slits [1]. These integrals also occur in the solution of quantum-mechanical diffraction and interference problems involving sharp-edged slits. In their classic text on path integrals in quantum mechanics, Feynman and Hibbs [2] have discussed the diffraction of the wave function of a non-relativistic particle, moving in one dimension, by a single slit. They showed that in the case of a Gaussian slit, the diffracted wave function is a spreading Gaussian wave packet that can easily be plotted. The diffracted

wave function for the more realistic sharp-edged slit, on the other hand, was shown to be expressible in terms of Fresnel integrals and tables of numerical values were used to plot the diffraction pattern. Kobe [3] has extended Feynman and Hibbs's treatment of a Gaussian slit to two such slits and has calculated the effect of the Aharonov-Bohm phase shift on the interference pattern when there is a magnetic flux string located between the two slits.

Modern computer algebra systems have made the evaluation, graphing and animation of many special mathematical functions almost routine operations. In this paper, Maple [4] will be used to plot diffraction and interference patterns that are associated with sharp-edged slits and hence involve combinations of Fresnel cosine and sine integrals. The Aharonov-Bohm effect will also be included. To illustrate how Maple can evaluate the Fresnel integral functions, the graphical representation of these functions known as the spiral of Cornu is reproduced in Section 2. Diffraction patterns due to a single slit and diffraction and interference patterns due to two slits are obtained in Sections 3 and 4, respectively. In the two-slit case, Maple code to generate an animation that exhibits the periodicity of the Aharonov-Bohm effect will be given. Finally, in Section 5, the classical limit is simulated in a Maple animation by allowing Planck's constant to approach zero.

2 Fresnel Integral Functions

The Fresnel cosine and sine integral functions are defined by

$$C(s) = \int_0^s \cos\left(\tfrac{1}{2}\pi v^2\right) dv \quad \text{and} \quad S(s) = \int_0^s \sin\left(\tfrac{1}{2}\pi v^2\right) dv. \tag{1}$$

It follows directly from the definitions that both C and S are odd functions. Also $C(s) \to \frac{1}{2}$ and $S(s) \to \frac{1}{2}$ as $s \to \infty$, as may be shown by contour integration [1]. The spiral of Cornu is the curve represented by the parametric equation

$$\mathbf{r}(s) = C(s)\mathbf{i} + S(s)\mathbf{j}, \quad -\infty < s < \infty. \tag{2}$$

Since $\mathbf{r}(0) = 0$ and $dC^2 + dS^2 = ds^2$, the parameter s is the signed arc length measured from the origin along the curve. Execution of the following Maple commands gives the portion of the spiral of Cornu shown in Figure 1.

```
> alias(C=FresnelC,S=FresnelS):
> plot([C(s),S(s),s=-4..4],scaling=constrained,labels=[C,S],
> labelfont=[HELVETICA,BOLD,14],numpoints=600);
```

The graph has traditionally been used as the basis of a method for estimating the values of $C(s)$ and $S(s)$ and is also useful [1] for deriving general properties of Fresnel diffraction patterns.



Figure 1: Spiral of Cornu

3 One-Slit Diffraction

Consider an idealized experiment in which a particle of mass m is emitted from the origin O of a rectangular Cartesian coordinate system $Oxyz$ when the time t is zero (see Figure 2). The motion parallel to the yz plane will be treated classically [3]; the y and z components of velocity will then be constant and the z component will be taken to be zero. At time T, the particle encounters a screen that is parallel to the xz plane and contains a long sharp-edged slit parallel to the z axis with width $2b$ and centre where $x = x_0$. It will be assumed that T and x_0 are both positive and that $0 < b < x_0$. When $0 < t < T$, the probability density for the x component of position is a uniform distribution and the corresponding wave function is not normalizable [2]. (This is because the particle was taken to be located at a point when $t = 0$.) The wave function is diffracted at the slit when $t = T$ and after a further time τ is given in terms of Fresnel integral functions by

$$\psi(x, T+\tau) = \sqrt{\frac{T}{4b(\tau+T)}} \exp\left[\frac{imx^2}{2\hbar\,(\tau+T)}\right] [C(u) + iS(u)]_{u_1}^{u_2} \qquad (3)$$

where

$$u_1 = \frac{x - (1+\tau/T)(x_0+b)}{\sqrt{(\pi\hbar/m)\tau(1+\tau/T)}}, \quad u_2 = \frac{x - (1+\tau/T)(x_0-b)}{\sqrt{(\pi\hbar/m)\tau(1+\tau/T)}} \qquad (4)$$

and $\hbar$ is Planck's constant divided by 2π. The wave function in Equation (3) is normalized to unity and the square of its modulus represents the conditional probability density (at time $T+\tau$) for the x component of position on a second screen parallel to the xz plane, given that the particle did not impinge on the first screen at time T.



Figure 2: Experimental Arrangement for One-Slit Diffraction

It is useful to introduce a dimensionless variable ξ and dimensionless parameters β, θ and μ through the definitions

$$\xi = \frac{x}{x_0}, \quad \beta = \frac{b}{x_0}, \quad \theta = 1 + \frac{\tau}{T} \quad \text{and} \quad \mu = x_0\sqrt{\frac{m}{\pi\hbar\tau\theta}}. \tag{5}$$

Then $\xi = 1$ at the slit centre and the width parameter β satisfies $0 < \beta < 1$. The parameter θ is a geometrical scaling factor that gives the ratio of the width of the slit in the first screen to the width of the projection of this slit from the origin O onto the second screen. It should be noted that the parameter μ varies directly as $\sqrt{m}$ and inversely as $\sqrt{\hbar}$. It follows from Equations (3), (4) and (5) that if

$$P(\xi;\beta,\theta,\mu) = \frac{1}{4\beta\theta}\left\{[C(\zeta_2) - C(\zeta_1)]^2 + [S(\zeta_2) - S(\zeta_1)]^2\right\}, \tag{6}$$

where

$$\zeta_1 = \mu\left[\xi - \theta(1+\beta)\right] \quad \text{and} \quad \zeta_2 = \mu\left[\xi - \theta(1-\beta)\right], \tag{7}$$

then

$$P(\xi;\beta,\theta,\mu)\,d\xi = |\psi(x, T+\tau)|^2\,dx. \tag{8}$$

For fixed values of the parameters β, θ and μ, the quantity $P(\xi;\beta,\theta,\mu)$ therefore represents the position probabilty density (in dimensionless form) at the point ξ.

To obtain graphs of the probability density, a Maple function `P` will be defined and then used with particular parameter values in the `plot` command. It will be convenient first to define Maple functions `CZ` and `SZ` that correspond to the combinations of Fresnel cosine and sine integrals appearing in Equation (6).

```
> CZ:=(xi,beta,theta,mu)->
> C(mu*(xi-theta*(1-beta)))-C(mu*(xi-theta*(1+beta))):
> SZ:=(xi,beta,theta,mu)->
> S(mu*(xi-theta*(1-beta)))-S(mu*(xi-theta*(1+beta))):
> P:=(xi,beta,theta,mu)->(1/(4*beta*theta))*
> (CZ(xi,beta,theta,mu)^2+SZ(xi,beta,theta,mu)^2):
> plot({P(xi,0.1,2.0,1.0),P(xi,0.5,2.0,1.0)},xi=-4..8,
> labels=[xi,P],labelfont=[HELVETICA,BOLD,14],
> numpoints=600,axes=frame);
```



Figure 3: One-Slit Diffraction Patterns

For both graphs shown in Figure 3, $\theta = 2.0$ and $\mu = 1.0$. The curve with the lower central peak corresponds to the narrower slit ($\beta = 0.1$), while the curve with the higher central peak corresponds to the wider slit ($\beta = 0.5$). There is in both cases a non-zero probability of detecting the particle outside the classically allowed region, which is the projection of the slit in the first screen from the origin onto the second screen. This region has centre where $\xi = \theta = 2.0$ and half width $\theta\beta$ (that is, 0.2 for $\beta = 0.1$ and 1.0 for $\beta = 0.5$). The quantum-mechanical broadening beyond the classical region is a manifestation of the diffraction of the wave function by the slit.

4 Two-Slit Diffraction and Interference

For a slit of width $2b$ with centre where $x = -x_0$, the diffracted wave function ϕ is obtained from Equations (3) and (4) by replacing x_0 by $-x_0$. When both slits are present, the complete wave function is a coherent superposition, with equal weightings, of the two one-slit wave functions ϕ and ψ. If the quantum-mechanical broadening is sufficiently large at time $T + \tau$ for the diffracted wave functions to overlap significantly, interference between the two components will be observable in the resulting probability density. Moreover, if the particle is electrically charged and there is a static magnetic flux string [3] embedded in the first screen parallel to and between the two slits, there will be a relative phase change between the two components which gives rise to a shift of the interference pattern on the second screen. This is the Aharonov-Bohm effect [5, 6, 3]. If $\alpha = q\Phi/(\hbar c)$, where q is the charge of the particle, Φ is the magnetic flux and c is the speed of light *in vacuo*, then the complete normalized wave function (apart from a multiplicative phase factor) at time $T + \tau$ is

$$\frac{1}{\sqrt{2}}\left[\phi(x, T+\tau) + e^{i\alpha}\psi(x, T+\tau)\right]. \tag{9}$$

The corresponding dimensionless probability density is given by

$$\begin{aligned} P(\xi;\alpha,\beta,\theta,\mu) = {} & \frac{1}{8\beta\theta}\times \\ & \Big(\{\cos\alpha\,[C(\zeta_2) - C(\zeta_1)] - \sin\alpha\,[S(\zeta_2) - S(\zeta_1)] + C(\eta_2) - C(\eta_1)\}^2 + \\ & \{\sin\alpha\,[C(\zeta_2) - C(\zeta_1)] + \cos\alpha\,[S(\zeta_2) - S(\zeta_1)] + S(\eta_2) - S(\eta_1)\}^2\Big) \end{aligned} \tag{10}$$

where ζ_1 and ζ_2 are defined as before by Equations (7) and

$$\eta_1 = \mu\,[\xi - \theta(-1+\beta)] \quad \text{and} \quad \eta_2 = \mu\,[\xi - \theta(-1-\beta)]. \tag{11}$$

The Maple function `P` is now redefined to reflect the presence of two slits and to incorporate the dimensionless flux parameter α. For this purpose, the previously defined functions `CZ` and `SZ` as well as two new functions `CE` and `SE` will be used.

```
> CE:=(xi,beta,theta,mu)->
> C(mu*(xi-theta*(-1-beta)))-C(mu*(xi-theta*(-1+beta))):
> SE:=(xi,beta,theta,mu)->
> S(mu*(xi-theta*(-1-beta)))-S(mu*(xi-theta*(-1+beta))):
> P:=(xi,alpha,beta,theta,mu)->(1/(8*beta*theta))*
> ((cos(alpha)*CZ(xi,beta,theta,mu)-
> sin(alpha)*SZ(xi,beta,theta,mu)+CE(xi,beta,theta,mu))^2+
> (sin(alpha)*CZ(xi,beta,theta,mu)+
> cos(alpha)*SZ(xi,beta,theta,mu)+SE(xi,beta,theta,mu))^2):
> plot(P(xi,0,0.1,2.0,1.0),xi=-10..10,numpoints=600,
> labels=[xi,P],labelfont=[HELVETICA,BOLD,14],axes=frame);
```



Figure 4: Two-Slit Diffraction and Interference Pattern

Figure 4 shows a typical two-slit diffraction and interference pattern. Because $\alpha = 0$ in this case, there is no Aharonov-Bohm effect and the graph is symmetric about the line $\xi = 0$.

It is evident from Equation (10) that the probability density is periodic in α with period 2π. The following Maple code generates an animation of the

Aharonov-Bohm effect in which the animation parameter α varies over one period in 8 steps.

```
> with(plots):
> n:=8:
> F:=plot(P(xi,0,0.1,2.0,1.0),xi=-3..3,numpoints=600,
> color=red):
> for i from 0 to n do
> G.i:=plot(P(xi,i*2*Pi/n,0.1,2.0,1.0),xi=-3..3,
> numpoints=600,color=blue):
> F.i:=display([F,G.i],labels=[xi,P],
> labelfont=[HELVETICA,BOLD,14]):
> od:
> display([seq(F.i,i=0..n)],insequence=true,axes=frame);
```

The periodicity of the effect may be demonstrated visually by playing the animation. (It should be emphasized that each frame of the animation shows the Aharonov-Bohm effect due to a *static* magnetic flux string.) The moving pattern (coloured blue) appears against a fixed background (coloured red) consisting of the central portion of Figure 4, for which $\alpha = 0$. As α changes, the moving pattern varies within the envelope of the one-parameter family of curves defined by Equation (10) with $\beta = 0.1$, $\theta = 2.0$ and $\mu = 1.0$. The third frame of the animation, corresponding to $\alpha = \frac{\pi}{2}$, is shown in Figure 5.



Figure 5: Aharonov-Bohm Effect with $\alpha = \frac{\pi}{2}$

5 Classical Limit

The classical limit of the quantum-mechanical probability density (10) is obtained by letting $\hbar$ tend to 0 or by letting m tend to ∞. In either case, the parameter μ defined in the third of Equations (5) tends to ∞ while the remaining parameters α, β and θ are fixed. The classical distribution is constant and non-zero within the region obtained by projecting the two slits in the first screen from the origin onto the second screen and is zero outside this region. The normalized classical distribution is therefore given by

$$\begin{aligned} P_{\mathrm{c}}(\xi;\beta,\theta) &= \frac{1}{4\beta\theta}\{H[\xi-\theta(-1-\beta)]-H[\xi-\theta(-1+\beta)]+ \\ &\qquad H[\xi-\theta(1-\beta)]-H[\xi-\theta(1+\beta)]\} \end{aligned} \tag{12}$$

where H is the Heaviside unit step function and $\xi \neq \theta(\pm 1 \pm \beta)$. Now as μ tends to ∞, each of the Fresnel integral functions appearing in the expression (10) for P tends to $\frac{1}{2}$ or $-\frac{1}{2}$ according as the argument ζ_1, ζ_2, η_1 or η_2 is positive or negative. By using this and considering separately the five open intervals obtained by removing the four points $\theta(\pm 1 \pm \beta)$ from the ξ axis, it may be shown that the classical distribution P_{c} is indeed the limit as μ tends to ∞ of the quantum-mechanical probability density P. (It may also be shown that at the four points where P_{c} is discontinuous, the limit of P is $1/(16\beta\theta)$.) The fact that the right-hand side of Equation (12) is independent of α then means that the Aharonov-Bohm effect vanishes in the classical limit.

An animation of the approach to the classical limit in which $\alpha = 0$, $\beta = 0.1$, $\theta = 2.0$ and μ increases exponentially is obtained as follows.

```
> n:=20:
> for i from 0 to n do
> F.i:=plot(P(xi,0,0.1,2.0,2^i),xi=-3..3,0..2,labels=[xi,P],
> labelfont=[HELVETICA,BOLD,14],numpoints=600):
> od:
> display([seq(F.i,i=0..n)],insequence=true);
```

The graph that appears in the final frame of the animation, which is shown in Figure 6, is a very good approximation to the graph of the classical distribution (12) with the stated parameter values. It should be noted, however, that as the final frame corresponds to a finite (though large) value of μ, Figure 6 is the graph of a function that is differentiable for all ξ.



Figure 6: Two-Slit Pattern with $\mu = 2^{20}$

References

[1] M. Born and E. Wolf, *Principles of Optics* (Pergamon, Oxford, 1975).

[2] R. P. Feynman and A. R. Hibbs, *Quantum Mechanics and Path Integrals* (McGraw-Hill, New York, 1965).

[3] D. H. Kobe, Ann. Phys. **123**, 381 (1979).

[4] Maple has been used previously to evaluate and plot combinations of Bessel functions and of confluent hypergeometric functions in connection with the bound-state Aharonov-Bohm effect. See W. P. Healy and H. Hung, in *Proceedings of the Third Asian Technology Conference in Mathematics*, W.-C. Yang *et al.* (editors), (Springer-Verlag, Singapore, 1998), p 300, and R. A. Samandra and W. P. Healy, J. Phys. A: Math. Gen. **31**, 9547 (1998). Many other applications of Maple in quantum mechanics may be found in M. Horbatsch, *Quantum Mechanics Using Maple* (Springer-Verlag, Berlin, 1995).

[5] Y. Aharonov and D. Bohm, Phys. Rev. **115**, 485 (1959).

[6] M. Peshkin and A. Tonomura, *The Aharonov-Bohm Effect* (Springer-Verlag, Berlin, 1989).